Math 173
Calculus I

Dr. Constant J. Goutziers
Department of Mathematics, Computer Science and Statistics
goutzicj@oneonta.edu

Lesson 8

Higher Order Derivatives

Initializations

```
> restart;
```

8.1 Higher order derivatives as Maple expressions.

Examples

Example 8.1.1

Compute the second and third order derivatives of $5x^2\cos 3x$ as a Maple expression.

Solution
Use the **diff** routine.

```
> e1:=5*x^2*cos(3*x);
```

$$e1 := 5\,x^2\cos(3\,x) \quad \textbf{(2.1.1.1)}$$

```
> der2:=diff(e1, x$2);
```

$$der2 := 10\cos(3\,x) - 60\,x\sin(3\,x) - 45\,x^2\cos(3\,x) \quad \textbf{(2.1.1.2)}$$

```
> der3:=diff(e1, x$3);
```

$$der3 := -90\sin(3\,x) - 270\,x\cos(3\,x) + 135\,x^2\sin(3\,x) \quad \textbf{(2.1.1.3)}$$

8.2 Higher order derivatives as Maple functions.

Examples

Example 8.2.1

Compute the second and third order derivatives of $f(x) = 5x^2\cos 3x$ as a Maple function.

Solution
This time, use the **D** routine.

```
> f:=x->5*x^2*cos(3*x);
```

$$f := x \to 5\,x^2\cos(3\,x) \quad \textbf{(3.1.1.1)}$$

```
> fpp:=(D@@2)(f);
```

$$fpp := x \to 10\cos(3\,x) - 60\,x\sin(3\,x) - 45\,x^2\cos(3\,x) \quad \textbf{(3.1.1.2)}$$

```
> fppp:=(D@@3)(f);
```

$$fppp := x \to -90\sin(3\,x) - 270\,x\cos(3\,x) + 135\,x^2\sin(3\,x) \quad \textbf{(3.1.1.3)}$$

8.3 Higher order derivatives of implicitly defined functions

Examples

Example 8.3.1

If $x^2y + xy^2 = 3$, compute $y''(x)$.

Solution

Use implicit differentiation. First compute $y'(x)$.

```
> e1:=x^2*y+x*y^2=3;
```

$$e1 := x^2\,y + x\,y^2 = 3 \qquad \textbf{(4.1.1.1)}$$

In order to inform the system that y is a function of x we replace y by $y(x)$.

```
> e2:=subs(y=y(x), e1);
```

$$e2 := x^2\,y(x) + x\,y(x)^2 = 3 \qquad \textbf{(4.1.1.2)}$$

Now take the derivative with respect to x of both sides of this equation.

```
> e3:=diff(e2, x);
```

$$e3 := 2\,x\,y(x) + x^2\left(\frac{d}{dx}\,y(x)\right) + y(x)^2 + 2\,x\,y(x)\left(\frac{d}{dx}\,y(x)\right) = 0 \qquad \textbf{(4.1.1.3)}$$

Solve for $\frac{dy}{dx}$.

```
> e4:=isolate(e3, diff(y(x), x));
```

$$e4 := \frac{d}{dx}\,y(x) = \frac{-2\,x\,y(x) - y(x)^2}{x^2 + 2\,x\,y(x)} \qquad \textbf{(4.1.1.4)}$$

```
> der1:=rhs(e4);
```

$$der1 := \frac{-2\,x\,y(x) - y(x)^2}{x^2 + 2\,x\,y(x)} \qquad \textbf{(4.1.1.5)}$$

Find $y''(x)$ by differentiating $y'(x)$ with respect to x.

```
> e5:=diff(der1, x);
```

$$e5 := \frac{-2\,y(x) - 2\,x\left(\frac{d}{dx}\,y(x)\right) - 2\,y(x)\left(\frac{d}{dx}\,y(x)\right)}{x^2 + 2\,x\,y(x)} - \frac{\left(-2\,x\,y(x) - y(x)^2\right)\left(2\,x + 2\,y(x) + 2\,x\left(\frac{d}{dx}\,y(x)\right)\right)}{\left(x^2 + 2\,x\,y(x)\right)^2} \qquad \textbf{(4.1.1.6)}$$

Use the formula for $\frac{dy}{dx}$ given in expression **e4**.

```
> e6:=subs(e4, e5);
```

$$e6 := \frac{-2\,y(x) - \frac{2\,x\left(-2\,x\,y(x) - y(x)^2\right)}{x^2 + 2\,x\,y(x)} - \frac{2\,y(x)\left(-2\,x\,y(x) - y(x)^2\right)}{x^2 + 2\,x\,y(x)}}{x^2 + 2\,x\,y(x)} \qquad \textbf{(4.1.1.7)}$$

$$-\frac{\left(-2\,x\,y(x)-y(x)^2\right)\left(2\,x+2\,y(x)+\frac{2\,x\left(-2\,x\,y(x)-y(x)^2\right)}{x^2+2\,x\,y(x)}\right)}{\left(x^2+2\,x\,y(x)\right)^2}$$

```
> der2:=simplify(e6);
```

$$der2 := \frac{6\,y(x)\left(x^2+x\,y(x)+y(x)^2\right)\left(x+y(x)\right)}{x^2\left(x+2\,y(x)\right)^3} \quad \textbf{(4.1.1.8)}$$

Make this formula easier to read, by replacing $y(x)$ by y.

```
> der2:=subs(y(x)=y, der2);
```

$$der2 := \frac{6\,y\left(x^2+x\,y+y^2\right)\left(x+y\right)}{x^2\left(x+2\,y\right)^3} \quad \textbf{(4.1.1.9)}$$

We check this result by using the **implicitdiff** command.

```
> a1:=factor(implicitdiff(e1, y, x$2));
```

$$a1 := \frac{6\,y\left(x^2+x\,y+y^2\right)\left(x+y\right)}{x^2\left(x+2\,y\right)^3} \quad \textbf{(4.1.1.10)}$$

```
>
```